
Stat 342  Lab Week 8 – Proc ttest, npar1way 

For this lab, you will need the cereal.csv dataset in the email.
Dataset source: Carl Schwarz's online notes
Textbook source: R and SAS Chapter 5, same as last week.
Additional Source: SAS/STAT(R) 9.2 User's Guide, Second Edition - Overview: TTEST Procedure

Part I, Importing, formatting, and exploratory analysis.
First, run the following blocks of code.

Step 1 – Importing the baseball batters dataset.

proc import
datafile='dir_location\cereal.csv'

out=cereal dbms=csv;
delimiter=',';
getnames=yes;

run;

Step 2 – Looking at the variables and data.
There are only 77 rows to this data, so we can look at all of the data at once.

proc contents data=cereal;
run;

proc print data=cereal;
run;

As with the other datasets
a. What do each of these variables represent?
b. What does a row of this data represent?
c. Is there are any missing data that you can see? Are there any impossible values. How might this 
affect further analysis. (Hint: Cups per serving)

Step 3 – Finding the distinct values for each category
Which of these variables is categorical? How many categories are in each one?
For a dataset this size, we can find this out manually, but it's good practice for larger datasets like the 
batting one. 

proc sql;
select distinct type from cereal;

proc sql;
select distinct manufacturer  from cereal;

proc sql;
select distinct shelf  from cereal; 



Part II, One and Two Sample T-Test

Step 1 – One-sample t-tests.

For a one sample t-test, the default null hypothesis of 0 pertains to the mean of an individual variable,  
so specify a mean for the null hypothesis with the h0 setting.

d. When more than one variable is selected in 'var', are the variables compared to each other, or 
independently to the given null hypothesis?

proc ttest data=cereal h0=10;
var fat sodium calories;

run;

You can only set one null hypothesis value, so even though we CAN set more than one response 
variable in proc ttest, it doesn't always make sense to.  A better strategy is to write a separate proc ttest 
for each response variable of interest.

Try doing a t-test for fat and sodium, and test each null against a 'greater than' alternative.

proc ttest data=cereal h0=50 sides=u alpha=0.05;
var sodium;

run;

proc ttest data=cereal h0=125 sides=u alpha=0.01;
var calories;

run;

Step 2 – Classical two-sample t-test.

To conduct a two-sample t-test, we need a class/by statement.
The variable that you use to determine which group each observation is in must have EXACTLY two 
distinct values. That means we can only do a two-sample t-test using the hot cereal vs. cold cereal var.

Also, the null hypothesis value refers to the difference in means between the groups. Typically a null 
hypothesis of zero difference makes sense, so we will use that, and not bother with an H0 = setting.

proc ttest data=cereal;
var calories;
class type; /* Hot or Cold cereal */

run;

What do you observe? Is there anything strange about either of the hot or cold categories? 
Assuming unequal variances, is there a significant difference between the hot and cold cereal calorie 
means?



Step 3 – Non-Parametric two-sample t-test

The classical two-sample t-test assumes that the values within each group are normally distributed. 
Sometimes that assumption is violated too much to ignore. In those cases, we can use non-parametric 
tests like the Wilcoxon Rank-Sum test to look for differences in the medians between distributions.

Non-Parametric versions of all one and two-sample t-tests are done with PROC NPAR1WAY
Wilcox test is one of several options here.

proc npar1way data=cereal wilcoxon;
var calories;
class type;

run;

Does the Wilcoxon rank-sum test detect a difference between medians in the calories between hot and 
cold cereal?

Part III, ANOVA (Analysis of Variance)

Step 1 – Parametric One-way ANOVA.

ANOVA is used in the place of a t-test when our grouping variable as three or more levels instead of 
just two. PROC ANOVA will tell us whether the differences between each of the group means are 
statistically significant. 

Like the t-test, ANOVA assumes that within each group that the values are normally distributed. It also 
assumes that the variances are equal.

Unlike the t-test procedure, PROC ANOVA is designed to handle a great variety of models. As such, 
we don't specify a response variable, we specify a model. Run a one-way anova on calories as a 
function of shelf.

proc anova data = cereal;
    class shelf;
    model calories = shelf; /* calories as a function of shelf */
run;

Are there significant differences in mean calories? How much of the variance in calories is explained 
by shelf? Repeat this analysis for sugar.

proc anova data = cereal;
    class shelf;
    model sugars = shelf; /* sugar as a function of shelf */
run;



Step 2 – Means comparisons

ANOVA can be used to make comparisons between individual pairs of groups, but adjustments for 
multiple comparisons are needed. The simplest of these are the Bonferroni and Sidak adjustments. A 
very popular method is the Tukey test.

proc anova data = cereal;
 class shelf;

model sugars = shelf; 
means shelf / bon sidak tukey;

run;

Is there evidence that the mean of a group is different from any other particular group? You will be able
to tell because there will be stars ** next to the pair of means for that group.

Conduct an anova on sugar content by manufacturer instead, which has 8 different groups instead of 3.

proc anova data = cereal;
 class manufacturer;

model sugars= manufacturer; 
means manufacturer / bon sidak tukey;

run;

Step 3 – Non-Parametric ANOVA

The npar1way procedure can be used in the place of the anova procedure if the normality assumption 
or the assumption of equal variance isn't valid.

We can test for equal variance, or Homogeneity Of Variance (HOV) with one of several HOV tests in 
the anova procedure.

proc anova data = cereal;
 class manufacturer;

model sugars= manufacturer; 
means manufacturer / hovtest=levene;

run;

The null hypothesis of the Levene test is that variance is equal across groups. Is there evidence that this
variances are different? (Hint: Use the Levene test p-value)

We could use the NPAR1WAY procedure if either assumption is violated. We can also do a classic 
anova with this procedure. This looks just like it does for t-tests.

proc npar1way data=cereal wilcoxon anova;
var sugars;
class manufacturer;

run;


